Old Dominion University Research Foundation

LANGLEY
GRANT
IN-36-CR
253620
47 P.

DEPARTMENT OF MATHEMATICAL SCIENCES
COLLEGE OF SCIENCES
OLD DOMINION UNIVERSITY
NORFOLK, VIRGINIA 23529

THEORETICAL STUDIES OF SOLAR LASERS AND CONVERTERS

By

John H. Heinbockel, Principal Investigator

Progress Report
For the period May 15, 1989 to December 31, 1989

Prepared for
National Aeronauatics and Space Administration
Langley Research Center
Hampton, Virginia 23665

Under
Research Contract NAG-1-757
Dr. Robert C. Costen, Technical Monitor
SSD-High Energy Science Branch

January 1990

DEPARTMENT OF MATHEMATICAL SCIENCES
COLLEGE OF SCIENCES
OLD DOMINION UNIVERSITY
NORFOLK, VIRGINIA 23529

THEORETICAL STUDIES OF SOLAR LASERS AND CONVERTERS

By

John H. Heinbockel, Principal Investigator

Progress Report
For the period May 15, 1989 to December 31, 1989

Prepared for
National Aeronautics and Space Administration
Langley Research Center
Hampton, Virginia 23665

Under
Research Grant NAG-1-757
Dr. Robert C. Costen, Technical Monitor
SSD-High Energy Science Branch

Submitted by the
Old Dominion University Research Foundation
P.O. Box 6369
Norfolk, Virginia 23508-0369

January 1990

THEORETICAL STUDIES OF SOLAR LASERS AND CONVERTERS

By

John H. Heinbockel*

Research for the period consisted of developing and refining the continuous flow laser model program. In addition, a working model for a two pass continuous wave amplifier was developed. The following is a summary of the mathematical development of a two pass amplifier for the n-C_3F_7I iodine laser. The geometry of the amplifier is illustrated in Figure 1. In this figure

P_{in} = power density into the amplifier (watts/cm^2),

W = transmission coefficient,

ϵ = radiation energy from I* (watt-sec),

C = speed of light,

P_{out} = output power density,

R_L = reflection coefficient for mirror,

$\rho+(Z)$ = photon density for wave motion in positive direction (cm^{-3}),

$\rho-(Z)$ = photon density for wave motion in negative direction (cm^{-3}).

The input power density P_{in} is assumed to be known. The photon density satisfies for all values Z

$$\rho+(Z)\ \rho-(Z) = K_0 \tag{1}$$

*Professor, Department of Mathematical Sciences, Old Dominion University, Norfolk, Virginia 23529.



Figure 1. Geometry for Two Pass Amplifier.

where K_0 is a constant. At Z = 0 we have

$$\rho+(0)\rho-(0) = K_0 \tag{2}$$

and at Z = L we have

$$\rho+(L)\ \rho-(L) = K_0. \tag{3}$$

The value of $\rho+(0)$ is determined from the assumed value for the input power density P_{in} and from this we calculate

$$\rho+(0) = \frac{P_{in}W}{\varepsilon\ C} \tag{4}$$

and consequently from the relation (2) we have

$$\rho-(0) = \frac{K_0}{\rho+(0)}\ .$$

In the above relations W is the transmission coefficient of the Brewster windows and K_0 is an assumed initial value. At Z = L we have $R_2 = W^2R_L$ as the reflection coefficient from the Brewster window. We then are able to calculate the return photon density from the Brewster window at Z = L. We find that this density should have the value

$$\rho_-(L) = R_2 \rho_+(L). \tag{5}$$

Combining the equations (2) (3) and (5) we determine that

$$R_2 \rho_+^2(L) = K_0 \qquad \text{or} \qquad \rho_+(L) = \sqrt{\frac{K_0}{R_2}} = \sqrt{\frac{\rho_+(0)\rho_-(0)}{R_2}} \tag{6}$$

Using the value (6) as the theoretical value for $\rho_+(L)$ at $Z = L$ we can compare this value with the calculate value for $\rho_+(L)$ obtained from an integration of the differential equations defining the chemical reactions occurring in the amplifier. We adjust the value for $\rho_-(0)$ until

$$\left| \rho_+(L)_{\text{theoretical}} - \rho_+(L)_{\text{calculated}} \right| < E \tag{7}$$

where E is an error criteria.

The above procedure is a shooting method for the determination of $\rho_-(0)$. When the error criteria is satisfied, the output power is obtainable from the relation

$$P_{out} = \rho_-(0)\ W\ \varepsilon\ C \tag{8}$$

and the gain is calculated from the relation

$$\text{gain} = 10 \log_{10}\left(\frac{P_{out}}{P_{in}}\right) \text{ (decibels)}. \tag{9}$$

The chemical reactions occurring in the laser tube are listed along with the values used for the rate constant. The assumed reactions are:

$RI + h\nu \xrightarrow{\xi_1} R + I^*$

$I_2 + h\nu \xrightarrow{\xi_2} I + I^*$ (51% of the time)

$I_2 + h\nu \xrightarrow{\xi_2} I + I$ (49% of the time)

Two Body Reactions

$I^* + R \xrightarrow{K_1} RI$, $K_1 = 1.0\ (10^{-14})$

$I + R \xrightarrow{K_2} RI$, $K_2 = 0.8\ (10^{-11})$

$R + R \xrightarrow{K_3} R_2$, $K_3 = 2.0\ (10^{-12})$

$R + RI \xrightarrow{K_4} R_2 + I$, $K_4 = 3.0\ (10^{-16})$

$R + I_2 \xrightarrow{K_5} RI + I$, $K_5 = 1.0\ (10^{-11})$

$R + RI \xrightarrow{K_6} R_2 + I^*$, $K_6 = 0.0$

$I^* + RI \xrightarrow{K_7} R + I_2$, $K_7 = 3.0(10^{-9})$

$I + RI \xrightarrow{K_8} I_2 + R$, $K_8 = 1.6(10^{-23})$

$I^* + RI \xrightarrow{Q_1} I + RI$, $Q_1 = 1.7(10^{-17})$

$I^* + I_2 \xrightarrow{Q_2} I + I_2$, $Q_2 = 3.8(10^{-11})e^{-4.4(10^{-3})(T-300)}$

$I^* + R \xrightarrow{Q_3} I + R$, $Q_3 = 3.7(10^{-18})$

$I^* + R_2 \xrightarrow{Q_4} I + R_2$, $Q_4 = 4.7(10^{-16})$

$I^* + I \xrightarrow{Q_5} I + I$, $Q_5 = 1.6(10^{-14})$

Three Body Reactions

$I^* + I + RI \xrightarrow{C_1} I_2 + RI$, $C_1 = 1.6(10^{-33})$

$I + I + RI \xrightarrow{C_2} I_2 + RI$, $C_2 = 5.7(0^{-33})e^{1310/\Gamma}$

$$I^*+I+I_2 \xrightarrow{C_3} I_2+I_2 \quad , \; C_3 = 0.0$$

$$I+I+I_2 \xrightarrow{C_4} I_2+I_2 \quad , \; C_4 = 10^{\alpha}$$

$$I+I+R_2 \xrightarrow{C_5} I_2+R_2 \quad , \; C_5 = 8.0(10^{-33})e^{1310/T}$$

$$I+I^*+R_2 \xrightarrow{C_6} I_2+R_2 \quad , \; C_6 = 0.0$$

where

$$\alpha = -29.439 - 5.844 \log_{10}\left(\frac{T}{300}\right) - 2.163\left[\log_{10}\frac{T}{300}\right]^2$$

Let B denote a scale factor and let y_i, i=1,7 denote the normalized values of concentration. Define the quantities:

$$X_1 = By_1 = [RI]$$

$$X_2 = By_2 = [R]$$

$$X_3 = By_3 = [R_2]$$

$$X_4 = By_4 = [I_2]$$

$$X_5 = By_5 = [I^*]$$

$$X_6 = By_6 = [I]$$

$$X_7 = By_7 = \rho+(Z)$$

$$X_8 = \rho-(Z) = K_0/X_7$$

$$\rho = \rho(Z) = X_7+X_8$$

where [ ] denotes concentration of substance, R is the iodide perflouride radical n-C_3F_7, I is iodine and I^* denotes the excited state of iodine. For ω the flow rate of n-C_3F_7I in the Z direction (cm/sec) and ρ the total photon density, the differential equations defining the chemical kinetics of the amplifier are given by:

$$\omega \frac{dy_1}{dz} = K_1By_2y_5+K_2By_2y_6-\xi_1y_1-K_4By_1y_2+K_5By_2y_4-K_7By_5y_1-K_6By_2y_1-K_8By_1y_6-y_1\frac{d\omega}{dz}$$

$$\omega \frac{dy_2}{dz} = \xi_1y_1-K_1By_2y_5-K_2By_2y_6-2K_3By_2^2-K_4By_1y_2-K_6By_1y_2-K_5By_2y_4$$

$$+K_7By_1y_5+K_8By_1y_6-y_2\frac{d\omega}{dz} - y_2/\tau_2$$

$$\omega \frac{dy_3}{dz} = K_3By_2{}^2+ K_6By_1y_2 + K_4By_1y_2-y_3\frac{d\omega}{dz} - y_3/\tau_3$$

$$\omega \frac{dy_4}{dz} = C_1B^2y_1y_5y_6+C_2B^2y_1y_6{}^2+C_3B^2y_4y_5y_6+C_4B^2y_4y_6{}^2-\xi_2y_4+K_7By_1y_5$$

$$- K_5By_2y_4+C_5B^2y_6{}^2y_3+K_8By_6y_1+C_6B^2y_6y_5y_3-y_4\frac{d\omega}{dz} - y_4/\tau_4$$

$$\omega \frac{dy_5}{dz} = \xi_1y_1+0.51\xi_2y_4-K_1By_2y_5-C_1B^2y_1y_5y_6-C_3B^2y_4y_5y_6-Q_1By_1y_5-Q_2By_4y_5$$

$$-C\sigma\cdot\rho\cdot(y_5-\frac{1}{2}y_6) + K_6By_2y_1-Q_3By_5y_2-Q_4By_5y_3-Q_5By_5y_6$$

$$-K_7By_5y_1-C_6B^2y_6y_5y_3-y_5\frac{d\omega}{dz} - y_5/\tau_5$$

$$\omega \frac{dy_6}{dz} = 1.49\ \xi_2y_4+ Q_1y_1y_5+Q_2By_4y_5-2C_5B^2y_6{}^2y_3-K_8By_6y_1+ C\sigma\rho\ (y_5-\frac{1}{2}y_6)$$

$$-C_1B^2y_1y_5y_6-2C_2B^2y_1y_6{}^2-C_3B^2y_4y_5y_6-2C_4B^2y_4y_6{}^2-K_2By_2y_6+K_4By_1y_2$$

$$+Q_3By_5y_2+Q_4By_5y_3+Q_5By_5y_6+K_5By_2y_4-C_6B^2y_6y_5y_3-y_6\frac{d\omega}{dz} - y_6/\tau_6$$

$$\frac{dy_7}{dz} = L_c y_7 (y_5 - \frac{1}{2} y_6) B\sigma$$

Normalizing the length of the amplifier by using the substitution $Z = SL$, $0 < S < 1$ the differential equations are solved numerically using a Runge-Kutta integration scheme. Appendix A contains the computer code for the simulation of a two pass continuous wave amplifier. Appendix B contains a representation of the output from the code. This Appendix contains figures illustrating how concentrations change with distance. Current research is directed toward the refining of this program and performing a systems study with the parameters involved in the code construction.

APPENDIX A

COMPUTER PROGRAM FOR SIMULATION OF TWO PASS CONTINUOUS WAVE AMPLIFIER

```
      PROGRAM TPAMP2(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT,TAPE8)
C
C
C        MAIN PROGRAM
C          TWO PASS CONTINUOUS WAVE AMPLIFIER
C
C           AMPPAR IS THE DATA FILE WITH NAMELIST VARIABLES DEFINED
C
C        COMPRESSIBLE FLOW LASER MODEL WHICH INCLUDES
C             EQUATION OF STATE
C             CONTINUITY EQUATION
C             MOMENTUM EQUATION
C             ENERGY EQUATION
C
C          TAPE8 IS FOR OUPUT DATA---RENAME AND SAVE TAPE8 IF YOU WANT
C                 TO SAVE THE OUTPUT DATA
C
C          PROBLEM IS SCALED IN THE AXIAL DIRECTION---LENGTH IS LC
C                      SCALED LENGTH IS ZERO TO ONE
C
C           SPECIFIC HEAT AT CONSTANT VOLUME IS FUNCTION OF TEMPERATURE
C                    OBTAINED FROM LEAST SQUARES FIT OF EMPIRICAL DATA
C                    FOR N-C3F7I IODINE LASER
C
C            USED IN PREDICTION MODE---GIVEN CONDITIONS AT Z=0.
C
C
C        THIS VERSION CONTAINS AN AUTOMATICALLY CHOSEN VARIABLE STEP
C        SIZE.  IT ALSO CONTAINS WALL EFFECT REACTIONS.(NOT USED THOUGH)
C
C         EFFECTS OF TEMPERATURE, FLOWRATE,DENSITY AND PRESSURE VARIATIONS
C         ARE CONSIDERED IN SUBROUTINES ARREN, PFLOW AND FLOW
C
C
        COMMON/BLK2/K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,C5
        COMMON/BLK27/Q1,Q2,Q3,Q4,Q5
        COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
        COMMON/BLK4/CHSI10,CHSI20,ABARO,Z1BAR,LC,XKO
        COMMON/BLK7/ABC,COO,CO,OMEG1,P,R1,R2,TM,XNRHO
         COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
         COMMON/BLK10/CF1,CF2,CF4,QFO,RSTAR,ZL,L,SF1,SF2,AR,AAO,BBO
        COMMON/BLK22/AD,V1,V2,GG
        COMMON/BLK23/WO,ETAO,PTO,FRAC
        COMMON/BLK28/TAU2,TAU4,TAU5,TAU6,TAU3,SIG
         COMMON/BLK299/TTT2,TTT3,TTT4,TTT5,TTT6
         REAL K1,K2,K3,K4,K5,K6,K7,K8
         REAL LC,L
C
C
         WRITE(6,123)
123      FORMAT(1X,20H START OF PROGRAM          )
```

```
C       DEFAULT VALUES
        IEND=0
        ICOUNT=0
        NG=0
        FRAC=0.0
         AR=2.
         TLE=39.0
         PTO=20.0
         WO=15.8
         ETAO=.38
        ZOL=7.50
         TO=300.
         AAO=147.23
C         AAO IN J/K MOL
C         BBO IN 1/K
         BBO=.0012
C        AAO,BBO USED IN LEAST SQ FIT OF SP.HEAT CONSTANT VOLUME
        LC=15.0
C         LC IN CM
         XNRHO=1.0
        RAD=0.0
        V1=0.0
        V2=0.0
         TTT2=1.0E18
         TTT3=1.0E18
         TTT4=1.0E18
         TTT5=1.0E18
         TTT6=1.0E18
        COO=.5E12
         R1=0.0
         R2=.975
        TO=300.
         PIN=1.0
         W=0.98
         ZE=5.0
        A=2.0
         TM= 1-R2
         OMEG1=5000.
C        OMEG1 IN CM/S
         CON=2.0E4
C
C
        NAMELIST/PARAM/PTO,OMEG1,CON,COO,R1,R2,TM,XNRHO,LC,ZOL,A,ZE,
       1 AR,IEND,FRAC,TO,RAD,V1,V2,TTT2,TTT4,TTT5,TTT6,TTT3,W,PIN
C
C
C         ASSUME PRESSURE PTO (TORR) AND TEMPERATURE TO(DEG K)
C         ARE GIVEN FOR LEFT END.  CALCULATE ETAO TO SATISFY GAS LAW
55       CONTINUE
         IF(IEND .EQ. 1) GO TO 600
         IF(NG.EQ.8) GO TO 600
```

```
      ICOUNT=ICOUNT+1
      READ(5,PARAM)
      IF(EOF(5))600,601
600   WRITE(6,603)
603   FORMAT(1X,28HEND OF FILE ENCOUNTERED-STOP)
      CALL PSEUDO
      DO 10 JJJ=1,9
      CALL GRAPHS(JJJ)
10    CONTINUE
      STOP 1313
C
C       W=TRANSMISSION COEFFICIENT
C       PIN=INPUT POWER DENSITY (WATTS/CM**2)
C
C       P=PRESSURE, TORR
C     AP=LASER BEAM DIAMETER CM
C       A=RADIUS OF TUBE (CM)
C       RAD=RADIUS OF TUBE WHERE CALCULATIONS ARE DONE.
C       T=TEMPERATURE DEG K
C       TO=TEMPERATURE LEFT END DEG K
C     PTO= PRESSURE TORR LEFT END
C     WO=FLOW RATE LEFT END M/SEC
C     ETAO=DENSITY OF GAS KG/M**3
C       OMEG1=FLOW RATE, CM/SEC , MAXIMUM FLOW RATE AT RAD=0
C      TAU2,TAU4,TAU5,TAU6 ARE LIFETIMES FOR RADIAL DIFFUSION OF
C     TTT2,TTT3,TTT4,TTT5,TTT6 ARE LIFETIME VALUES FOR TAU
C      THE QUANTITIES [R],[I2],[I*] AND [I].
C      TAU3 IS DIFFUSION COEFFICIENT FOR [R2]
C       CON=PEAK CONCENTRATION  , SOLAR CONSTANTS
C       COO=INITIAL GUESS AT RHO-PLUS AT ZERO
C           WHICH IS SQUARE OF (COO*R1)
C       R1= REFLECTIVITY AT LEFT END
C       R2= REFLECTIVITY AT RIGHT END
C      ZE=DISTANCE FOR LIGHT INTENSITY TO DIMINISH BY FACTOR 1/E
C       TM= TRANSMISSION COEFFICIENT (OUTPUT MIRROR)
C      ZOL=POINT ALONG AXIS WHERE MAXIMUM ILLUMINATION OCCURS
C     IN THE CASE ILLUMINATION IS A BELL SHAPED CURVE
C      IN THE CASE OF A SQUARE WAVE, 2*ZOL IS CUT OFF POINT
C        THE POINT 2*ZOL IS WHERE ILLUMINATION BEGINS TO DIMINISH
C      LC=LENGTH OF LASER CAVITY IN CM
C      S= SCALED LENGTH WHICH GOES FROM ZERO TO ONE
C         LENGTH IS SCALED FROM ZERO TO ONE
C     ZL=2*ZOL    LENGTH WHICH IS ILLUMINATED
C      FRAC=FRACTION OF PEAK CONCENTRATION WHICH GOES
C            INTO HEAT
C           DEFAULT VALUE IS FRAC=0.024 I.E. 2.40 PERCENT
C      H IS STEP SIZE FOR NUMERICAL INTEGRATION OVER SCALED LENGTH
C
C
601   CONTINUE
      P=PTO
      T=TO
      ETAO=P*(1.01325E5)*296./(8314.3*760.*T)
C     PTO IS PRESSURE IN TORR
C     ETAO IS DENSITY IN KG/M**3
C     ETAO IN KG/M**3
```

```
C       P IN TORR
C       T IN DEG K
        CMIN=1.0E4
        CMAX=1.0E20
       ZL=2*ZOL
        L=LC
        C0=CON
        C11=CON
       CALL COEFFS
C       W0 IN M/SEC
        W0=OMEGA/100.
C       OMEG1 AND OMEGA ARE IN CM/SEC
        WRITE(6,198)
198     FORMAT(///)
       WRITE(6,199)   ZE,ZOL,CON,OMEGA,FRAC,COO,R1,R2,P,T
199     FORMAT(1X, T5,5HZE = ,F10.3,T30,6HZOL = ,E15.7,T55,6HCON = ,
      1 E15.7,T80,9HOMEGA = ,E15.7,T104,7HFRAC = ,F9.4,/,
      2 1X,T5,6HCOO = ,E15.7, T30,6H R1 = ,F10.7, T55,6H R2 = ,F10.7,
      3 T81,4H P = ,E15.7 ,T105,7HTEMP =  ,F10.3)
C      SET UP COEFFICIENTS IN DIFFERENTIAL EQUATIONS
C        SET PRINTER OFF
C        IPRINT=0       PRINTER OFF
C        IPRINT=1       PRINTER ON
        IPRINT=0
C       SET STEP SIZE H=.025 UNITS
        H=0.025
C      INTEGRATE DIFFERENTIAL EQUATIONS FROM S=0 TO S=1.0
       X1=COO
       CALL INTEG(IPRINT,H)
C
C          INTERVAL HALVING SCHEME
C
       Y1=ABC
        IF(Y1.LT.0) PER=5.0
CC        IF(Y1.LT.0) PER=1.1
        IF(Y1.GT.0) PER=.1
CC        IF(Y1.GT.0) PER=.9
702     CONTINUE
C       IN THIS VERSION COO IS ASSUMED VALUE FOR RHO-(0)
        COO=(PER)*COO
        IF(COO .LT. CMIN) STOP 5432
        IF(COO .GT. CMAX) STOP 2345
       X2=COO
       CALL INTEG(IPRINT,H)
       Y2=ABC
C       ABC IS XKO INITIAL MINUS XKO CALCULATED (PERCENT ERROR)
        IF((Y1*Y2).LT. 0)GO  TO 701
        X1=COO
        Y1=Y2
        GO TO 702
 701    CONTINUE
        W1=.6
        W2=.4
       COO=W2*X1+W1*X2
 708   CONTINUE
        CALL INTEG(IPRINT,H)
```

```
        X3=COO
        Y3=ABC
       IF(ABS(Y3).LT.0.05) GO TO 555
 704    CONTINUE
        IF((Y1*Y3).LT. 0) GO TO 705
 C      Y1 & Y3 ARE OF THE SAME SIGN
        X1=X3
        Y1=Y3
        COO=.5*(X1+X2)
        GO TO 708
 705    CONTINUE
 C      Y1 & Y3 ARE OF OPPOSITE SIGN
        X2=X3
        Y2=Y3
        COO=.5*(X1+X2)
        GO TO 708
 555     IPRINT=1
         H=0.025
         CALL INTEG(IPRINT,H)
         GO TO 55
       END
```

ORIGINAL PAGE IS
OF POOR QUALITY

```
        SUBROUTINE GRAPHS(JJJ)
 C
 C
 C      AS THE NAME SUGGESTS THIS SUBROUTINE PLOTS GRAPHS OF THE
 C           NUMERICAL OUTPUT
 C      MULTIPLE PLOTS ON EACH GRAPH ---UP TO MAXIMUM OF EIGHT
 C
 C
        COMMON/BLK4/CHSI10,CHSI20,ABARO,Z1BAR,LC,XKO
        COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
        COMMON/BLK30/DATA(50,50),NDMAX,FLRATE(8),FDPTDAT(50,40)
        DIMENSION X(50),Y(50),YY(50,8)
        REAL LC
 C
 C
 C      JJJ=1,9
 C      JJJ=1 PLOT R VS Z
 C      JJJ=2 PLOT I2 VS Z
 C      JJJ=3 PLOT I* VS Z
 C      JJJ=4 PLOT I VS Z
 C      JJJ=5 PLOT INV VS Z
 C      JJJ=6 PLOT FLOWRATE VS Z
 C      JJJ=7 PLOT DENSITY VS Z
 C      JJJ=8 PLOT PRESSURE VS Z
 C      JJJ=9 PLOT TEMPERATURE VS Z
 C
 C
 C       DATA ARRAY IS BY COLUMNS
 C       Z,R,I2,I*,I,INV,Z,R,I2,I*,I,INV,...
 C       ND=NUMBER OF DATA POINTS,=NUMBER OF ROWS IN DATA ARRAY
 C       NG=NUMBER OF CURVES PER GRAPH
```

```
C        IC IS CODE TO DETERMINE NUMBER OF GRAPHS TO PLOT
C        IC=0 FOR MORE THAN ONE GRAPH
C        IC=1 FOR LAST GRAPH (USED FOR ONLY ONE GRAPH)
C
C
         IC=0
         NLAST=NG-1
        SXC=5.0
        SYC=5.0
        SYCI=7.0
C       SXC=LENGTH OF XSCALE
C       SYC=LENGTH OF YSCALE
C       SYCI=LENGTH OF YSCALE FOR INVERSION
         IF(NLAST .EQ. 0) IC=1
C        PLOT JJJ VS Z
         IF(JJJ .GE. 6) GO TO 700
         DO 10 I=1,NDMAX
         X(I)=DATA(I,1)
         DO 20 J=1,NG
         NN=(J-1)*6
         NCOL=NN+1+JJJ
        YY(I,J)=DATA(I,NCOL)
20      CONTINUE
10      CONTINUE
C
C       FIND YMAX,YMIN
C
C
        DMAX=0.0
       YMAX=0.0
       YMIN=0.0
        ZMIN=0.0
C
C        FIND YMAX AND YMIN
C
        DO 70 I=1,NDMAX
        DO 71 J=1,NG
        IF( YY(I,J) .GT. YMAX) YMAX=YY(I,J)
        IF( YY(I,J) .LT. YMIN) YMIN=YY(I,J)
71      CONTINUE
70      CONTINUE
C
C       PLOT FIRST DATA CURVE
C
C        JJJ=1
        ZMAX=1.00
        DO 40 I=1,NDMAX
        Y(I)=YY(I,1)
40     CONTINUE
        LOGMAX=1+ALOG10(ABS(YMAX))
        POWER1=10.**LOGMAX
        TEST=YMAX/POWER1
        YMAXX=0.025
        IF(TEST .GT. 0.025) YMAXX=.05
        IF(TEST .GT. 0.05) YMAXX=0.10
       IF(TEST .GT. 0.10) YMAXX=0.25
```

```
      IF(TEST .GT. 0.25) YMAXX=0.50
       IF(TEST .GT. 0.5) YMAXX=1.00
       YMAX=YMAXX*POWER1
       IF(JJJ.NE. 5) GO TO 507
506    CONTINUE
      IF(YMIN .GE. 0) GO TO 507
      IF(YMIN .LT. 0.0) LOGMAX2=1+ALOG10(ABS(YMIN))
       POWER=10.**LOGMAX2
       TEST=ABS(YMIN)/POWER
       YMINN=0.001
       IF(TEST .GT. 0.001) YMINN=0.005
       IF(TEST .GT. 0.005) YMINN=0.01
       IF(TEST .GT. 0.01) YMINN=0.05
       IF(TEST .GT. 0.05) YMINN=0.1
       IF(TEST .GT. 0.10) YMINN=0.5
       IF(TEST .GT. 0.5) YMINN=1.0
       YMIN=-YMINN*POWER
       SYC=SYCI
507    CONTINUE
       ZMIN=0.0
       IF(JJJ.GT.1) GO TO 50
C      JJJ=1   PLOT R VS Z
C
       CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1  22,22HZ, SCALE DISTANCE,          ,
     2  11,11H    [C3F7] ,0,
     3  SXC,SYC,0.75,0.75)
       GO TO 600
50     CONTINUE
       IF(JJJ.GT.2) GO TO 100
C
C      JJJ=2   PLOT I2 VS Z
C
       CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1  23,23HZ, SCALE DISTANCE,         ,
     2  9,9H    [I2]  ,0,
     3  SXC,SYC,0.75,0.75)
      GO TO 600
100   CONTINUE
      IF(JJJ.GT.3) GO TO 200
C
C     JJJ=3   PLOT I* VS Z
C
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 23,23HZ, SCALE DISTANCE,       ,
     2  9,9H    [I*]  ,0,
     3  SXC,SYC,0.75,0.75)
      GO TO 600
200   CONTINUE
      IF(JJJ.GT.4) GO TO 300
C
C     JJJ=4    PLOT  I VS Z
C
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1  23,23HZ, SCALE DISTANCE,       ,
     2  8,8H    [I]  ,0,
```

```
      3  SXC,SYC,0.75,0.75)
       GO TO 600
300    CONTINUE
C
C      JJJ=5   PLOT INV VS Z
C
       CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
      1  23,23HZ, SCALE DISTANCE,        ,
      2  24,24H             [I*]-.5*[I]    ,0,
      3  SXC,SYC,0.75,0.75)
600    CONTINUE
C      PLOT REST OF CURVES OR EXIT IF ONLY ONE CURVE
C
        NLAST=NG-1
        IF(NG .EQ. 2) GO TO 5001
        IF(NLAST .EQ. 0) GO TO 601
        DO 500  J=2,NLAST
        DO 501 I=1,NDMAX
        Y(I)=YY(I,J)
501     CONTINUE
        CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
      1  1,1H   ,1,1H    , 0,SXC,SYC,0.75,0.75)
500     CONTINUE
5001      CONTINUE
C
C       PLOT LAST CURVE
C
        DO 60 I=1,NDMAX
       Y(I)=YY(I,NG)
60     CONTINUE
       CALL INFOPLT(1,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
      1 1,1H  ,1,1H  ,0,SXC,SYC,0.75,0.75)
601       CONTINUE
       CALL NFRAME
       RETURN
700    CONTINUE
C      GET DATA TO PLOT
       DO 701 I=1,NDMAX
       X(I)=FDPTDAT(I,1)
       DO 702 J=1,NG
       NN=JJJ-4+(J-1)*5
       YY(I,J)=FDPTDAT(I,NN)
702    CONTINUE
C       FDPTDAT IS STORED IN THE FORM:
C        Z,FLOWRATE,DENSITY,PRESSURE,TEMPERATURE,Z,FLOWRATE,...
701    CONTINUE
C      FIND MAX AND MIN FOR Y VALUES
       YMIN=100.
       YMAX=0.0
       DO 703 I=1,NDMAX
       DO 704 J=1,NG
       IF(YY(I,J) .LT. YMIN) YMIN=YY(I,J)
       IF(YY(I,J) .GT. YMAX) YMAX=YY(I,J)
704    CONTINUE
703    CONTINUE
       MAX=10.*YMAX+1.0
```

ORIGINAL PAGE IS
OF POOR QUALITY

```
       YMAX=FLOAT(MAX)/10.
       MIN=10.*YMIN-1.0
       YMIN=FLOAT(MIN)/10.
      ZMIN=0.0
      ZMAX=1.0
C     PLOT FIRST CURVEC
      DO 705 I=1,NDMAX
      NN=1
705   Y(I)=YY(I,NN)
C     PLOT JJJ VS Z
      IF(JJJ .GT. 6) GO TO 800
C     JJJ=6, PLOT FLOWRATE VS Z
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 25,25HZ, AXIAL DISTANCE, SCALED ,
     2 15,15HFLOWRATE CM/SEC     , 0 ,
     3 10.,5.,0.75,0.75)
      GO TO 820
800   CONTINUE
      IF(JJJ .GT. 7) GO TO 801
C     JJJ=7, PLOT DENSITY VS Z
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 25,25HZ, AXIAL DISTANCE, SCALED ,
     2 16,16HDENSITY, KG/M**3     ,0,
     3 10.,5.,0.75,0.75)
      GO TO 820
801   CONTINUE
      IF(JJJ .GT. 8) GO TO 802
C     JJJ=8, PLOT PRESSURE VS Z
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 25,25HZ, AXIAL DISTANCE, SCALED ,
     2 14,14HPRESSURE, TORR     , 0 ,
     3 10.,5.,0.75,0.75 )
      GO TO 820
802   CONTINUE
C     JJJ=9, PLOT TEMPERATURE VS Z
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 25,25HZ, AXIAL DISTANCE, SCALED ,
     2 18,18HTEMPERATURE, DEG C     , 0 ,
     3 10.,5.,0.75,0.75 )
820   CONTINUE
      IF(NLAST .EQ. 0) GO TO 901
C      PLOT REST OF THE CURVES
       IF(NG .EQ. 2) GO TO 5031
      DO 503 J=2,NLAST
      DO 504 I=1,NDMAX
      NN=J
      Y(I)=YY(I,NN)
504   CONTINUE
      CALL INFOPLT(IC,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 1,1H  ,1,1H  ,0,10.,5.,0.75,0.75 )
503   CONTINUE
5031     CONTINUE
C     PLOT LAST CURVE
      DO 505 I=1,NDMAX
      Y(I)=YY(I,NG)
505   CONTINUE
```

```
      CALL INFOPLT(1,NDMAX,X,1,Y,1,ZMIN,ZMAX,YMIN,YMAX,1.0,
     1 1,1H  ,1,1H  ,0,10.,5.,0.75,0.75 )
901   CONTINUE
      CALL NFRAME
      RETURN
      END




       SUBROUTINE PFLOW
C      SUBROUTINE TO CALCULATE THE PARAMETERS FOR SUBROUTINE FLOW
C      PARAMETERS STORE IN COMMON BLK10
      COMMON/BLK7/ABC,COO,CO,OMEG1,P,R1,R2,TM,XNRHO
      COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
       COMMON/BLK10/CF1,CF2,CF4,QFO,RSTAR,ZL,L,SF1,SF2,AR,AAO,BBO
      COMMON/BLK23/WO,ETAO,PTO,FRAC
       COMMON/BLK29/ ZZZ,TZZ,PZZ,ETAZZ,WZZ
      REAL L
C
C      AR IS BEAM DIAMETER RADIUS IN CM
       SF1=1000./296.
C       SF1 =1/(M*1.E-03), WHERE M IS THE MOLECULAR WEIGHT
C      CVS (J/K KG) =SF1*CV (J/K MOL)
       SF2=(1.01325E5)/760.
C       SF2 EQUALS 133.3 PA/TORR
C      SF1,SF2 ARE SCALE FACTORS FOR THE CORRECT UNITS OF
C      P=PRESSURE (N/M**2) , SF1 (MOLE/KG)
C      PT=PRESSURE (TORR),   SF2 (N/M**2)/TORR
C      TEST TO SEE IF WO (M/SEC) IS NEAR MAX VALUE OF 5.30*SQRT(TO)
      TEST=5.3*SQRT(TO)
      TEST1=ABS(100.*(WO-TEST)/TEST)
      IF(TEST1 .LT. 10.0) WRITE(6,778)
778   FORMAT(1X,52HWARNING WO VALUE IS WITHIN 10 PERCENT OF ITS MAXIMUM
     1 ,/,1X,35HALLOWABLE VALUE OF 5.3*SQRT(TO)                        )
C      ZL=LIGHT SOURCE LENGTH IN CM
C      ETA=DENSITY (KG/M**3)
C      L=TUBE LENGTH IN CM
C      RSTAR=GAS CONSTANT (JOULE/KG DEG K)
C      T=TEMPERATURE (DEG K)
C      CV=SPECIFIC HEAT AT CONSTANT VOLUME
C       RSTAR (J/K KG) = SF1*R(J/K MOL)
C      WO,WL=FLOW VELOCITY (M/SEC) (SUBSCRIPTS O,L FOR START,END)
       PO=SF2*PTO
       RSTAR=8314.3/296.0
       CF1=ETAO*WO
       CF2=CF1*WO+PO
       QFO=FRAC*(1.40E3)*CO *2./(A*1.E-2)
C      QFO IN W/M**3
C     INPUT (W/M**2) DISTRIBUTED OVER VOLUME (2(PI)A)/PI(A**2) IN M2/M3
       WZZ=WO
       RETURN
       END


       SUBROUTINE FLOW(Z,T,PTORR,WSS,ETA,DWDZ)
```

```
      COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
C     SUBROUTINE TO CALCULATE T,P,WSS,ETA AS FUNCTION OF Z
      COMMON/BLK10/CF1,CF2,CF4,QFO,RSTAR,ZL,L,SF1,SF2,AR,AAO,BBO
      COMMON/BLK23/WO,ETAO,PTO,FRAC
C
      COMMON/BLK29/ ZZZ,TZZ,PZZ,ETAZZ,WZZ
     REAL L
      ICOUNT=0
      Q=QFO*Z/100.
C     ENERGY INPUT TERM DETERMINED BY QFO
C     Z IS IN CM AND QFO IS IN W/M**3
      IF(Z.GT.ZL)     Q=QFO*ZL/100.
     XXX=BBO*(TO-300.)
     CALL ETO(XXX,EEO)
     WSS=WZZ
50    CONTINUE
     T=((CF2/CF1)*WSS-WSS*WSS)/RSTAR
     DTDW=((CF2/CF1)-2.*WSS)/RSTAR
     XXX=BBO*(T-300.)
     CALL ETO(XXX,EE1)
     F=RSTAR*(T-TO)+(AAO/BBO)*SF1*(EE1-EEO)+
    1 .5*(WSS*WSS-WO*WO)-Q/CF1
     FP=(RSTAR+SF1*AAO*EE1)*DTDW+WSS
     W1=WSS-F/FP
     ERR=ABS(100.*(W1-WSS)/WSS)
     IF(ERR .LT. .25) GO TO 100
      IF(ICOUNT .GT. 95) WRITE(6,357)ICOUNT,Z,WZZ,WSS,CF1,CF2,TO,Q,FM
357   FORMAT(1X,I5,1X,(3E16.7,/))
     WSS=W1
      ICOUNT=ICOUNT+1
      IF(ICOUNT .GT. 100) STOP 4444
     GO TO 50
100  CONTINUE
     WSS=W1
     T=((CF2/CF1)*WSS-WSS*WSS)/RSTAR
     TC=T-273
      ETA=CF1/WSS
      PNM2=ETA*RSTAR*T
      PTORR=PNM2/SF2
     ZZZ=Z
     TZZ=TC
     PZZ=PTORR
     ETAZZ=ETA
     WZZ=WSS
     XXX=BBO*(T-300.)
     CALL ETO(XXX,CVO)
     CV=CVO*AAO
     CVS=SF1*CV
     XNUM=RSTAR*QFO
     XDEN=(CF2-2.*CF1*WSS)*(CVS+RSTAR)+RSTAR*CF1*WSS
     DWDZ=XNUM/XDEN
      RETURN
      END
```

```
      SUBROUTINE ARREN(TEMP)
C     SUBROUTINE FOR ARRENHIUS EXPRESSION OF RATE COEFFICIENTS
C     BASIC ASSUMPTIONS
C      FOR QI TERMS QI=QIO*EXP(-BETA*(TEMP-TO))
C      TREAT KI TERMS LIKE CI TERMS
C
     COMMON/BLK2/K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,C5
     COMMON/BLK27/Q1,Q2,Q3,Q4,Q5
     COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
      COMMON/BLK11/KK1,KK2,KK3,KK4,KK5,KK6,KK7,KK8
      COMMON/BLK12/QQ1,QQ2,QQ3,QQ4,QQ5
     COMMON/BLK13/CC1,CC2,CC3,CC4,CC5,CC6
      REAL K1,K2,K3,K4,K5,K6,K7,K8,KK1,KK2,KK3,KK4,KK5,KK6,KK7,KK8
C
C      REFERENCE J.S. COHEN AND O.P. JUDD
C                J.APPL. PHYS., VOL 55, NO. 7, APRIL 1984
C      COEFFICIENTS MODIFIED TO ACHIEVE SPECIFIC VALUES AT TEMPERATURE
C      OF 276 DEGREES K.
C
       BETA=4.4E-3
       SF1=1.0
      XXX=-BETA*(TEMP-300)
      CALL ETO(XXX,YYY)
       SF2=YYY
       K1=KK1*SF1
       K2=KK2*SF1
       K3=KK3*SF1
       K4=KK4*SF1
       K5=KK5*SF1
       K6=KK6*SF1
       K7=KK7*SF1
       K8=KK8*SF1
     C1=CC1*SF1
     C2=CC2*EXP(1360.00/TEMP)
     C3=CC3*SF1
      XYZ=-29.437-5.844*ALOG10(TEMP/300.)+2.163*(ALOG10(TEMP/300.))**2
      C4=CC4**XYZ
      C5=CC5*EXP(1310.000/TEMP)
     C6=CC6*SF1
      Q1=QQ1*SF1
      Q2=QQ2*EXP((-4.4E-3)*(TEMP-300.))
      Q3=QQ3*SF1
      Q4=QQ4*SF1
      Q5=QQ5*SF1
      RETURN
      END



     SUBROUTINE ETO(X,Y)
C    NEGATIVE EXPONENTIAL FUNCTION
     IF(X .LT. -670.) GO TO 100
```

ORIGINAL PAGE IS
OF POOR QUALITY

```
      Y=EXP(X)
      RETURN
100   Y=0.
      RETURN
      END


      FUNCTION CHSI1(Z)
C       Z1BAR IS CUTOFF POINT OF ILLUMINATION
C       ABARO IS FRONT CUTOFF POINT OF ILLUMINATION
C        CHSI1 IS A  CONSTANT
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
      COMMON/BLK4/CHSI10,CHSI20,ABARO,Z1BAR,LC,XKO
      COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
      REAL LC
      IF(Z.LT.ABARO) GO TO 100
      IF(Z.LT.Z1BAR) GO TO 200
C     Z GREATER THAN Z1BAR
100   CHSI1=0.0
C      CHSI1 HAS UNITS OF SEC^-1
      RETURN
200   CONTINUE
      CHSI1=CHSI10
      RETURN
      END


      FUNCTION CHSI2(Z)
C        ABARO IS FRONT CUTOFF POINT FOR ILLUMINATION
C        Z1BAR IS CUTOFF POINT FOR ILLUMINATION
C        CHSI2 IS A CONSTANT
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
      COMMON/BLK4/CHSI10,CHSI20,ABARO,Z1BAR,LC,XKO
      COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
      REAL LC
      IF(Z.LT.ABARO) GO TO 100
      IF(Z.LT.Z1BAR) GO TO 200
C     Z GREATER THAN Z1BAR
CC     XXX=-(Z-Z1BAR)/ZE
CC     CALL ETO(XXX,YYY)
CC    CHSI2=CHSI20*YYY
C     CHSI2 HAS UNITS OF SEC^-1
      RETURN
100   CHSI2=0.0
      RETURN
200   CONTINUE
      CHSI2=CHSI20
      RETURN
      END
```

```
      SUBROUTINE COEFFS
C       THIS SUBROUTINE DEFINES THE COEFFICIENTS IN THE DIFFERENTIAL
C       EQUATIONS TO BE SOLVED.
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
      COMMON/BLK22/AD,V1,V2,GG
      COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
      COMMON/BLK4/CHSI10,CHSI20,ABARO,Z1BAR,LC,XKO
      COMMON/BLK7/ABC,C00,C0,OMEG1,P,R1,R2,TM,XNRHO
       COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
       COMMON/BLK11/KK1,KK2,KK3,KK4,KK5,KK6,KK7,KK8
       COMMON/BLK12/QQ1,QQ2,QQ3,QQ4,QQ5
      COMMON/BLK13/CC1,CC2,CC3,CC4,CC5,CC6
      COMMON/BLK28/TAU2,TAU4,TAU5,TAU6,TAU3,SIG
       COMMON/BLK299/TTT2,TTT3,TTT4,TTT5,TTT6
      REAL KK1,KK2,KK3,KK4,KK5,KK6,KK7,KK8,LC
C
C     COEFFICIENTS IN THE DIFFERENTIAL EQUATIONS
C
      OMEGA=OMEG1
C       OMEGA=OMEG1 IS FLOW RATE IN CM/SEC
C       OMEGA IS FLOWRATE AT RADIUS R=RAD
C        SEE SUBROUTINE VELOC
C
       CALL VELOC(OMEG1,RAD,OMEGA,A)
C
      ABARO=0.0
C     ABARO IS START OF ILLUMINATION
C     Z1BAR=2*ZOL IS THE POINT ON AXIS WHERE ILLUMINATION BEGINS TO DIMINISH
       CHSI10=(3.04E-3)*C0*XNRHO
C     C00 IS THE CONCENTRATION IN SOLAR CONSTANTS
       CHSI20=(3.38E-2)*C0/2.5
       Z1BAR=2*ZOL
       EPSNU=1.5E-19
C      WATTS*SEC
       A00=2.0E17
       B00=0.443
      TAU2=TTT2
      TAU3=TTT3
      TAU4=TTT4
      TAU5=TTT5
      TAU6=TTT6
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C        (CHSI1)     HV + RI --> R + I*
C        (CHSI2)     HV + I2 --> I* + I     (51% OF THE TIME)
C        (CHSI2)     HV + I2 -->  I + I    (49% OF THE TIME)
C
C          (KK1)      I* + R --> RI
C          (KK2)      I + R --> RI
C          (KK3)      R + R --> R2
C          (KK4)      R + RI --> R2 + I
C          (KK5)      R + I2 --> RI + I
C          (KK6)      R + RI --> R2 + I*
C          (KK8)      I* + RI --> I2 + R
C
```

```
C            (QQ1)      I* + RI --> I + RI
C            (QQ2)      I* + I2 --> I + I2
C            (QQ3)      I* + R --> I + R
C            (QQ4)      I* + R2 --> I + R2
C            (QQ5)      I* + I --> I + I
C
CC            (CC1)      I* + I + RI --> I2 + R1
CC            (CC2)      I + I + RI --> I2 + RI
CC            (CC3)      I* + I + I2 --> I2 + I2
CC            (CC4)      I + I + I2 --> I2 + I2
CC            (CC5)      I + I + R2 --> I2 + R2
CC            (CC6)      I + I* + R2 --> I2 + R2
C
C
C        WALL REACTIONS
C
C          (V1)      I + I + WALL --> I2 + WALL
C          (V2)     R2 + I + WALL --> R + RI + WALL
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C
C
C   WILSON ET AL. (1984) AND STOCK ET AL. (1986)
C      KK1 = 5.6E-13
C   COHEN AND JUDD (1984)
C      KK1 = .9E-13
C   VINOKKUROV AND ZALESSKKII (1979)
       KK1 = 1.E-14
C
C
C   WILSON ET AL. (1984) AND STOCK ET AL. (1986)
C      KK2 = 2.3E-11
C   BREDERLOW ET AL. (1983)
       KK2 = .8E-11
C   COHEN AND JUDD (1984)
C       KK2 = .7E-11
C
C
C
C   WILSON ET AL. (1984) AND STOCK ET AL. (1986)
C      KK3 = 2.6E-12

C   BREDERLOW ET AL. (1983) AND COHEN AND JUDD (1984)
       KK3 = 2.0E-12
C
C
C   WILSON ET AL. (1984), STOCK ET AL. (1986), AND COHEN AND JUDD (1984)
       KK4 = 3.E-16
C
C
C   COHEN AND JUDD (1984)
       KK5 = 1.0E-11
C
C   WILSON ET AL. (1984) AND STOCK ET AL. (1986)
C      KK6 = 3.2E-17
```

```
C  BREDERLOW ET AL. (1983) AND COHEN AND JUDD (1984)
      KK6 = 0.
C
C
C  COHEN AND JUDD (1984)
      KK7 = 3.0E-19
C
C
C
C  COHEN AND JUDD (1984)
      KK8 = 1.6E-23
C
C
C  WILSON ET AL. (1984) AND STOCK ET AL. (1986)
C     QQ1 = 2.0E-16
C  COHEN AND JUDD (1984)
C     QQ1 = 1.7E-16
C  BREDERLOW ET AL. (1983)
      QQ1 = 1.7E-17
C
C
C
C  COHEN AND JUDD (1984)
C     QQ2 = 3.80E-11*EXP(-4.4E-3*(T-300.))
      QQ2=3.8E-11
C      SUBROUTINE ARREN ADDS TEMPERATURE EFFECT
C  BREDERLOW ET AL. (1983)
C     QQ2 = 3.0E-11
C  WILSON ET AL. (1984) AND STOCK ET AL. (1986)
C     QQ2 = 1.9E-11
C
C
C
C  COHEN AND JUDD (1984)
      QQ3 = 3.7E-18
C
C
C  COHEN AND JUDD (1984)
      QQ4 = 4.7E-16
C
C
C  COHEN AND JUDD (1984)
      QQ5 = 1.6E-14
C
C
C
C  COHEN AND JUDD (1984) (A DIFFERENT REACTION, PERHAPS)
CC     CC1 = 1.E-32
CC  WILSON ET AL. (1984) AND STOCK ET AL. (1986)
CC     CC1 = 3.2E-33
CC  HOHLA AND KOMPA (1976)
      CC1 = 1.6E-33
CC
CC
C  COHEN AND JUDD (1984)
C     CC2 = 5.7E-33*EXP(1360./T)
```

```
      CC2 = 5.7E-33
C     SUBROUTINE ARREN ADDS TEMPERATURE EFFECT
CC  HOHLA AND KOMPA (1976)
CC       CC2 = 2.1E-33*EXP(1600./T)
C  BREDERLOW ET AL. (1983)
CC       CC2 = 3.8E-31
C  WILSON ET AL. (1984) AND STOCK ET AL. (1986)
CC       CC2 = 8.5E-32
C
C
C  WILSON ET AL. (1984) AND STOCK ET AL. (1986)
CC       CC3 = 8.E-32
C  BREDERLOW ET AL. (1983) AND COHEN AND JUDD (1984)
      CC3 = 0.
C
C
C  WILSON ET AL. (1984) AND STOCK ET AL. (1986)
CC       CC4 = 3.8E-30
C  COHEN AND JUDD (1984)
C     CC4 = 10.0**(-29.437-5.844*ALOG10(T/300.)
      CC4 = 10.0
C     SUBROUTINE ARREN ADDS TEMPERATURE EFFECT
C  BREDERLOW ET AL. (1983)
CC       CC4 = 2.9E-30
C
C
C  COHEN AND JUDD (1984)
C     CC5 = 8.E-33*EXP(1310./T)
      CC5 = 8.0E-33
C     SUBROUTINE ARREN ADDS TEMPERATURE EFFECT
C
C
C  BREDERLOW ET AL. (1983) AND COHEN AND JUDD (1984)
      CC6 = 0.0
C
C
C     V1 = 0.0
C
C
C     V2 = 0.0
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C
C     KK1=.903E-13
C     KK2=80.E-12
C     KK3=.65E-12
C     KK4=1.000E-16
C     KK5=1.089E-11
C     KK7=.1517E-18
C     KK8=1.6E-23
C
C     QQ1=.476E-16
C     QQ2=1.9E-11
C     QQ3=.1235E-17
C     QQ4=1.57E-16
```

```
C      QQ5=.53E-14
C
CC      CC1=1.053E-33
CC      CC2=45.0E-32
CC      CC3=.4447E-31
CC      CC4=4.94E-30
CC      CC5=3.6E-31
CC      CC6=      1.8E-32
C
C      V1=       1.0E-12
C      V2=       1.0E-11
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C
       GG=2*(.18/LC)**2
       C=3.0E10
CC      B=P*(3.5E16)
       B=(9.66E18)*P/TO
       B2=B*B
       B3=B2*B
       RETURN
       END



        SUBROUTINE VELOC(OMEG1,RAD,OMEGA,A)
C       VELOCITY PROFILE IS ASSUMED TO BE PARABOLIC
C        MAX VELOCITY AT RAD=0.0
C       ZERO VELOCITY AT RAD=A
C       A IS RADIUS OF TUBE
C        OMEG1 IS MAXIMUM VELOCITY ALONG CENTERLINE
C
C       CALCULATE VELOCITY OMEGA AT R=RAD
C            O .LE. RAD  .LE. A
C
C       TYPE OF FLOW
        OMEGA=(OMEG1/(A*A))*(RAD-A)**2
        RETURN
        END



        SUBROUTINE PRINT(Z,TEMP)
       COMMON/BLK2/K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,C5
       COMMON/BLK27/Q1,Q2,Q3,Q4,Q5
       COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
       COMMON/BLK4/CHSI10,CHSI20,ABAR0,Z1BAR,LC,XK0
       COMMON/BLK7/ABC,C00,C0,OMEG1,P,R1,R2,TM,XNRHO
       COMMON/BLK22/AD,V1,V2,GG
```

```
      COMMON/BLK28/TAU2,TAU4,TAU5,TAU6,TAU3,SIG
       COMMON/BLK299/TTT2,TTT3,TTT4,TTT5,TTT6
      REAL K1,K2,K3,K4,K5,K6,K7,K8,LC
C
       IF(Z .EQ. 0.0) WRITE(6,10)
10     FORMAT(1X,20HCOEFFICIENTS AT Z=0         )
       IF(Z .GE. 1.0) WRITE(6,11)
11     FORMAT(1X,20HCOEFFICIENTS AT Z=L         )
C       WRITE OUT COEFFICIENTS
       WRITE(6,100) K1,K7,Q1,C5,TTT2
       WRITE(6,101) K2,K8,Q2,V1,TTT3
       WRITE(6,102) K3,C1,Q3,V2,TTT4
       WRITE(6,103) K4,C2,Q4,SIG,TTT5
       WRITE(6,104) K5,C3,Q5,TEMP,TTT6
       WRITE(6,105) K6,C4,C6,LC
100    FORMAT(T5,5HKK1 = ,E15.7,T30,5HKK7 = ,E15.7,T60,5HQQ1 = ,E15.7 ,
     1 T85,5HCC5 =   , E15.7   , T103,7HTTT2 = ,E15.7 )
101    FORMAT(T5,5HKK2 = ,E15.7,T30,5HKK8 = ,E15.7,T60,5HQQ2 = ,E15.7 ,
     1 T85,5HV1 =   , E15.7   , T103,7HTTT3 = ,E15.7 )
102    FORMAT(T5,5HKK3 = ,E15.7,T30,5HCC1 = ,E15.7,T60,5HQQ3 = ,E15.7 ,
     1  T85, 5HV2 =   , E15.7 , T103,7HTTT4 = ,E15.7 )
103    FORMAT(T5,5HKK4 = ,E15.7,T30,5HCC2 = ,E15.7,T60,5HQQ4 = ,E15.7 ,
     1 T85, 8HSIGMA = , E15.7   , T103,7HTTT5 = ,E15.7 )
104    FORMAT(T5,5HKK5 = ,E15.7,T30,5HCC3 = ,E15.7,T60,5HQQ5 = ,E15.7 ,
     1 T85,7HTEMP =   ,F9.2   , T103,7HTTT6 = ,E15.7 )
105    FORMAT(T5,5HKK6 = ,E15.7,T30,5HCC4 = ,E15.7,T60,5HCC6 = ,E15.7 ,
     1 T85, 4HL =   ,F9.2 )
       RETURN
       END


      SUBROUTINE FUN(N,S,Y,F)
C       THIS SUBROUTINE DEFINES THE RIGHT HAND SIDE
C       OF THE DIFFERENTIAL EQUATIONS FOR THE CHEMICAL KINETICS
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
      DIMENSION Y(7),F(7)
      COMMON/BLK1/X7,POWER
      EXTERNAL CHSI1,CHSI2
      COMMON/BLK2/K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,C5
      COMMON/BLK27/Q1,Q2,Q3,Q4,Q5
      COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
      COMMON/BLK4/CHSI10,CHSI20,ABAR0,Z1BAR,LC,XK0
      COMMON/BLK7/ABC,C00,C0,OMEG1,P,R1,R2,TM,XNRHO
      COMMON/BLK22/AD,V1,V2,GG
      COMMON/BLK28/TAU2,TAU4,TAU5,TAU6,TAU3,SIG
       COMMON/BLK299/TTT2,TTT3,TTT4,TTT5,TTT6
      REAL K1,K2,K3,K4,K5,K6,K7,K8,LC
C
C     QY=QUANTUM YIELD
      QY=1.0
C
C       F(I),I=1,6 ARE RATES OF CHANGES FOR THE CONCENTRATIONS
C        F(1)=D[RI]/DZ                     F(2)=D[R]/DZ
C        F(3)=D[R2]/DZ                     F(4)=D[I2]/DZ
```

```
C          F(5)=D[I*]/DZ                     F(6)=D[I]/DZ
C
20      CONTINUE
         Z=LC*S
C        S IS SCALED LENGTH VARIABLE
C        O  .LE. S .LE. 1
C         FLOW CALCULATES TEMPERATURE, PRESSURE, FLOWRATE
C         AND DENSITY RESPONSE AS FUNCTION OF Z
C
C        Z IS DISTANCE IN CM
C        CALCULATE GAS PARAMETERS AS FUNCTION OF Z
C        CALL FLOW(Z,TEMP,PRESS,FLOWR,DENSITY,DWDZ)
         CALL FLOW(Z,TEMP,PRESS,FLOWR,DENSITY,DWDZ)
C        WRITE(6,357)Z,TEMP,PRESS,FLOWR,DENSITY,DWDZ
357      FORMAT(1X,3E16.7,/,3E16.7 )
C        TEMP IS TEMPERATURE DEG K
C        PRESS IS PRESSURE IN TORR
C        FLOWR IS FLOWRATE IN M/SEC
C        DENSITY IS GAS DENSITY IN KG/M**3
         OMEGA=FLOWR*100.
C        OMEGA IS FLOW RATE IN CM/SEC
C        CALCULATE COEFFICIENTS AS FUNCTION OF TEMP AND Z
         CALL ARREN(TEMP)
         TAU5=(TTT5)*PRESS
         TAU6=TAU5
C
C
C        CONSTANTS COME VIA COMMON BLKS 2 AND 3
C        K'S IN CM**3/SEC
C        C'S IN CM**6/SEC
C        Q'S IN CM**3/SEC
C        RBWEROISYIN)WBCM**2
         X7STAR=Y(7)*B+X8
         DIF=Y(5)-.5*Y(6)
         CALL SIGMA(SIG2)
         SIG=SIG2
       F(1)=K1*B*Y(2)*Y(5)+K2*B*Y(2)*Y(6)-CHSI1(Z)*Y(1)-K4*B*Y(1)*Y(2)
      1 +K5*B*Y(2)*Y(4)  -K7*B*Y(5)*Y(1)-K6*B*Y(2)*Y(1)   +V2*B*Y(3)*Y(6)
      2  -K8*B*Y(6)*Y(1)  -Y(1)*DWDZ
       F(2)=CHSI1(Z)*Y(1)-K1*B*Y(2)*Y(5)-K2*B*Y(2)*Y(6)-2*K3*B*Y(2)*Y(2)
      1 -K4*B*Y(1)*Y(2)-K6*B*Y(1)*Y(2)-K5*B*Y(2)*Y(4)+V2*B*Y(3)*Y(6)
      2 +K7*B*Y(5)*Y(1)  +K8*B*Y(6)*Y(1)  -Y(2)*DWDZ-Y(2)/TAU2
       F(3)=K3*B*Y(2)*Y(2)+K6*B*Y(1)*Y(2)+K4*B*Y(1)*Y(2)-V2*B*Y(3)*Y(6)-
      1 Y(3)*DWDZ  -Y(3)/TAU3
       A1=C1*B2*Y(1)*Y(5)*Y(6)+C2*B2*Y(1)*Y(6)*Y(6)+C3*B2*Y(4)*Y(5)*Y(6)
       A2=C4*B2*Y(4)*Y(6)*Y(6)-CHSI2(Z)*Y(4)+K7*B*Y(5)*Y(1)
      1 -K5*B*Y(2)*Y(4)  +V1*B*Y(6)*Y(6)  +C5*B2*Y(6)*Y(6)*Y(3)
       F(4)=A1+A2+K8*B*Y(6)*Y(1)+C6*B2*Y(6)*Y(5)*Y(3)-Y(4)*DWDZ-Y(4)/TAU4
       A3=QY*CHSI1(Z)*Y(1)+0.51*CHSI2(Z)*Y(4)-K1*B*Y(2)*Y(5)
       A4=-C1*B2*Y(1)*Y(5)*Y(6)-C3*B2*Y(4)*Y(5)*Y(6)-Q1*B*Y(1)*Y(5)
       A5=-Q2*B*Y(4)*Y(5)-C*SIG*X7STAR*DIF +K6*B*Y(2)*Y(1)
       F(5)=A3+A4+A5-Q3*B*Y(5)*Y(2)-Q4*B*Y(5)*Y(3)-Q5*B*Y(5)*Y(6)
      1 -K7*B*Y(5)*Y(1)-C6*B2*Y(6)*Y(5)*Y(3)-Y(5)*DWDZ-Y(5)/TAU5
       A6=1.49*CHSI2(Z)*Y(4)+Q1*B*Y(1)*Y(5)+Q2*B*Y(4)*Y(5)
      1  -2*C5*B2*Y(6)*Y(6)*Y(3)   -K8*B*Y(6)*Y(1)
       A7=C*SIG*X7STAR*DIF -C1*B2*Y(1)*Y(5)*Y(6)
```

```
      A8=-2*C2*B2*Y(1)*Y(6)*Y(6)-C3*B2*Y(4)*Y(5)*Y(6)
      A9=-2*C4*B2*Y(4)*Y(6)*Y(6)-K2*B*Y(2)*Y(6)+K4*B*Y(1)*Y(2)
       A10=Q3*B*Y(5)*Y(2)+Q4*B*Y(5)*Y(3)+Q5*B*Y(5)*Y(6)
     1 +K5*B*Y(2)*Y(4)   -V2*B*Y(3)*Y(6)  -2*V1*B*Y(6)*Y(6)
      F(6)=A6+A7+A8+A9+A10-C6*B2*Y(6)*Y(5)*Y(3)  -Y(6)*DWDZ-Y(6)/TAU6
C        SCALED EQUATIONS IN THE Z-DIRECTION
      DO 10 I=1,6
10    F(I)=LC*F(I)/OMEGA
       F(7)=LC*Y(7)*DIF*B*SIG
      RETURN
      END


       SUBROUTINE SIGMA(SIG)
C      THIS SUBROUTINE DEFINES THE CROSS SECTION SIGMA
      COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
      COMMON/BLK7/ABC,C00,C0,OMEG1,P,R1,R2,TM,XNRHO
       REAL NU,NUS,NU0,NU1,NU2,NU3,NU4 ,NU5
       PI=3.14159
       PIS=PI*PI
       NU=C/1.315246E-4
       NUS=NU*NU
       PISNUS=PIS*NUS*4.
       G=0
       CS=C*C
       NU0=NU
       NU1=NU0+.141*C
       NU2=NU1+.068*C
       NU3=NU0-.427*C
       NU4=NU3-.026*C
       NU5=NU4-.068*C
       DELTA23=NU-NU5
       DELTA22=NU-NU4
       DELTA21=NU-NU3
       DELTA34=NU-NU0
       DELTA33=NU-NU1
       DELTA32=NU-NU2
       TEMP0=293
       TWALL=TEMP0
       T1=TWALL
       A=5.434
       A1=A*2.4/7.7*CS
       A2=A*3.0/7.7*CS
       A3=A*2.3/7.7*CS
       A4=A*5.0/7.7*CS
       A5=A*2.2/7.7*CS
       A6=A*0.6/7.7*CS
       FUGTEMP=SQRT(T1/300.)
       ALPHAM=1.88E7*FUGTEMP
       DELDOP=2.51E8*FUGTEMP
       DELNU=DELDOP+ALPHAM*P
       SIGMA23=A1/(PISNUS*DELNU)/(1+(2.*DELTA23/DELNU)**2)*5./12.
       SIGMA22=A2/(PISNUS*DELNU)/(1+(2.*DELTA22/DELNU)**2)*5./12.
       SIGMA21=A3/(PISNUS*DELNU)/(1+(2.*DELTA21/DELNU)**2)*5./12.
```

```
      SIGMA34=A4/(PISNUS*DELNU)/(1+(2.*DELTA34/DELNU)**2)*7./12.
      SIGMA33=A5/(PISNUS*DELNU)/(1+(2.*DELTA33/DELNU)**2)*7./12.
      SIGMA32=A6/(PISNUS*DELNU)/(1+(2.*DELTA32/DELNU)**2)*7./12.
      SIGMAT=SIGMA23+SIGMA22+SIGMA21+SIGMA34+SIGMA33+SIGMA32
      SIG=SIGMAT
      RETURN
      END



      SUBROUTINE INTEG(IPRINT,H)
C      THIS SUBROUTINE INTEGRATES THE SYSTEM OF DIFFERENTIAL EQUATIONS
C      DEFINING THE CHEMICAL KINETICS OF N-C3F7I IODINE LASER
C
C      USING A VARIABLE STEP SIZE 7TH ORDER RUNGE KUTTA-FEHLBERG METHOD.
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
      DIMENSION YO(7),X(7),WK(84)
      COMMON/BLK1/X7,POWER
      COMMON/BLK3/B,B2,B3,C,A00,B00,EPSNU,OMEGA,C6
      COMMON/BLK4/CHSI10,CHSI20,ABARO,Z1BAR,LC,XKO
      COMMON/BLK7/ABC,C00,CO,OMEG1,P,R1,R2,TM,XNRHO
       COMMON/BLK8/ZOL,ZE,NG,TO,RAD,A,PIN,W
      COMMON/BLK22/AD,V1,V2,GG
      COMMON/BLK23/WO,ETAO,PTO,FRAC
      COMMON/BLK29/ZZZ,TZZ,PZZ,ETAZZ,WZZ
       COMMON/BLK30/DATA(50,50),NDMAX,FLRATE(8),FDPTDAT(50,40)
      EXTERNAL FUN,CHSI1,CHSI2
      REAL LC
C
C     INTEGRATE SYSTEM FROM S=0 TO S=1.0 USING RUNGE-KUTTA METHOD
C
C     X(1)=RI
C     X(2)=R
C     X(3)=R2
C     X(4)=I2
C     X(5)=I*
C     X(6)=I
C     X(7)=RHO+
C     X8=RHO-
C     X9=I*-.5*I
C
C     INITIALIZE CONSTANTS FOR FLOW EQUATIONS
C     SEE COMMON BLK10 FOR THESE CONSTANTS--NEEDED FOR SUB FLOW
      CALL PFLOW
      ND=0
C       TEST FOR PRINT CONDITIONS
       IF(IPRINT.EQ. 0) GO TO 229
       NG=NG+1
C      NG IS THE NUMBER OF GRAPHS TO BE SAVED
C     MAXIMUM NG=8
C     FLOWRATE(NG) IS LABEL FOR DATA SAVED
C     SEE ALSO PROGRAM PLOTD--WHICH CAN PLOT THE SAVED DATA
C
      FLRATE(NG)=OMEGA
       WRITE(8,331) LC
```

```
331      FORMAT(1X,11H   LC     =  ,F10.2)
229       CONTINUE
55       CONTINUE
       N=7
       TOL=1.0E-5
       PD=1.0
       MTH=1
       WZZ=WO
C       H IS STEP SIZE IN SCALED UNITS BETWEEN PRINT OUTS
C       IPRINT=0 OFF, IPRINT=1 ON
        HMIN=1.0E-9
        HMAX=H/100.
        HUSE=HMIN*1000.
        IERR=0
C      INITIAL CONDITIONS
       Z0=0.0
       YO(1)=1.0
        Z1=0.0
       DO 9 I=2,6
9      YO(I)=0.0
C        INITIALIZE FLOW,DENSITY,PRESSURE AND TEMPERATURE AT Z=0
        CALL FLOW(Z0,TEMP,PRESS,FLOWR,DENSITY,DWDZ)
C        GUESS AT INITIAL CONDITIONS FOR X(8)
       X70=PIN*W/(EPSNU*C)
C      X70 IS RHO+(0) WHICH IS GIVEN
C      INITIAL CONDITION FOR RHO-(0) IS UNKNOWN--ASSUME COO VALUE
       XKO=X70*COO
       X8=XKO/X70
        YO(7)=X70/B
        IF(IPRINT .EQ. 0) GO TO  300
        CALL FUN(N,Z0,YO,F)
        TEMP=TZZ+273.
        IF(Z0.EQ.0.0)CALL PRINT(Z0,TEMP)
        WRITE(6,191)
191    FORMAT(///,T7,1HZ,T20,4H[RI],T32,4H[R],T45,5H[R2],T57,4H[I2],
      1 T69,4H[I*],T80,4H[I] ,T91,6H[RHO+],T103,6H[RHO-],T112,
      2 9HINVERSION        )
300    CONTINUE
       DO 10 I=1,7
10     X(I)=B*YO(I)
C
       X8=XKO/X(7)
       X9=X(5)-.5*X(6)
       X7STAR=X(7)+X8
C        USE SUBROUTINE SIGMA TO CALCULATE CROSS SECTION SIGMA
       CALL SIGMA(SIG2)
        IF(IPRINT .EQ. 0) GO TO 222
       WRITE(6,199)Z0,(X(I),I=1,7),X8,X9
        WRITE(6,303)ZZZ,TZZ,PZZ,ETAZZ,WZZ
        CALL FUN(N,Z0,YO,F)
        TEMP=TZZ+273.
        IF(Z0.EQ.1.0)CALL PRINT(Z0,TEMP)
222      CONTINUE
C       SAVE THE DATA FOR FUTURE PLOT ROUTINES
C          DATA ARRAYS ARE DATA(50,50) AND FDPTDAT(50,40)
C
```

```
          IF(IPRINT .EQ. 0) GO TO 227
          ICOL=(NG-1)*6
          ND=ND+1
          DATA(ND,ICOL+2)=X(2)
          DATA(ND,ICOL+1)=ZO
          DATA(ND,ICOL+3)=X(4)
          DATA(ND,ICOL+4)=X(5)
          DATA(ND,ICOL+5)=X(6)
          DATA(ND,ICOL+6)=X9
          NNN=(NG-1)*5
          FDPTDAT(ND,NNN+1)=ZO
          FDPTDAT(ND,NNN+2)=WZZ
          FDPTDAT(ND,NNN+3)=ETAZZ
          FDPTDAT(ND,NNN+4)=PZZ
          FDPTDAT(ND,NNN+5)=TZZ
C         WRITE Z,R,I2,I*,I,INV
          WRITE(8,6773) ZO,X(2),X(4),X(5),X(6),X9
         WRITE(8,6773)ZO,WZZ,ETAZZ,PZZ,TZZ
6773      FORMAT(1X,F6.3,2X,5(2X,E15.6))
227       CONTINUE
         IF(ZO.LE.0.0) GO TO 3567
          IF(IPRINT .EQ. 0) GO TO 223
          WRITE(6,303)ZZZ,TZZ,PZZ,ETAZZ,WZZ
223         CONTINUE
303       FORMAT(1X,T2,3HZ= ,F10.3,2X,T15,3HT= ,F7.3,2X,T30,
        1 7HPTORR=  ,F9.4,2X,
        2 T55,9HDENSITY = ,F9.6,2X,T80,3HW=  ,E14.7    )
3567       CONTINUE
199      FORMAT(1X,E12.5,8E12.5,E12.5 ,E12.5  )
C
C
C          USE 7TH ORDER RUNGE KUTTA INTEGRATION SCHEME WITH VARIABLE STEP
C          STEP SIZE CAN VARY FROM HMIN TO HMAX
C          ZO IS STARTING VALUE FOR Z
C          Z1 IS NEXT STOPPING POINT IN INTEGRATION SCHEME
C          TOL IS TOLERENCE
C          IERR IS ERROR CODE TO DETERMINE IF INTEGRATION WAS SUCCESSFUL
C
C
100      CONTINUE
          Z1=Z1+H
          IF(Z1 .GT. 1.00) GO TO 111
          CALL RKF7(N,ZO,Z1,YO,TOL,FUN,PD,MTH,HMIN,HMAX,HUSE,WK,IERR)
          IF(IERR .NE. 0)  WRITE(6,444)  IERR,ZO,Z1,(YO(I),I=1,7)
         IF(IERR .NE. 0) STOP 1717
444      FORMAT(1X,18HIERR IS NOT ZERO    ,I5, (4E16.7,/)             )
200      CONTINUE
          X(7)=B*YO(7)
          X8=XKO/X(7)
60       FORMAT(1X,5(2X,E14.6))
         GO TO 300
500      CONTINUE
          DO 110 I=1,7
110       X(I)=B*YO(I)
          X8=XKO/X(7)
           X9=X(5)-.5*X(6)
```

```
111      CONTINUE
         XX7L=X(7)
C        X(7) IS CALCULATED RHO+(L) VALUE
         XKCAL=R2*X(7)*X(7)
C        XKCAL IS CALCULATED VALUE FOR XKO
C        ABC IS DIFFERENCE BETWEEN CALCULATED AND INITIAL VALUE
         DIF=((XKO-XKCAL)/XKCAL)*100.
         ABC=DIF
         IF(IPRINT .EQ. 0) GO TO 224
         WRITE(6,202)DIF,XKCAL,XKO,COO
224      CONTINUE
202      FORMAT(1X,13HDIFFERENCE = ,E18.9,2X,12HXKCAL    =     ,E18.9,
       1 2X,10HXKO    =     ,E18.9,2X,6HCOO =   ,E18.9 )
237      CONTINUE
C
C        RHOMIN=RHO-(0)=XKO/RHO+(0)=XKO/X70
         RHOMIN=XKO/X70
         POUT=RHOMIN*W*EPSNU*C
         GAINDB=10.*ALOG10(POUT/PIN)
         GAIN=POUT/PIN
         IF(IPRINT .EQ. 0) GO TO 226
         WRITE(6,193)R1,R2,P,PIN,POUT,GAINDB,GAIN
226      CONTINUE
193      FORMAT(1X,5HR1 = ,F10.7,2X,5HR2 = ,F10.7,1X,
       1 8HPRESS = ,F10.7,2X,6HPIN = ,E14.7,3X,7HPOUT = ,E14.7,/,
       2 1X,9HGAINDB =  ,E14.7,3X,7HGAIN = ,E14.7  )
501      CONTINUE
         NDMAX=ND
C        NDMAX IS THE MAXIMUM NUMBER OF DATA POINTS
         RETURN
         END
```

APPENDIX B

OUTPUT FROM SIMULATION MODEL

INPUT PARAMETERS

PTO = 15.0

OMEG1 = 3000

CON = 1689

COO = 2,13398E8

R1 = 0.00

R2 = 0.9584792

LC = 1000

ZOL = 500

XNRHO = 1.0

TO = 300.0

OUTPUT

P_{in}	P_{out}	Gain	Gain db
10.0	128.4570	12.8457	11.0875
5.0	124.9000	24.9801	13.97595
2.0	122.2330	61.1166	17.8616
1.0	119.9611	119.9611	20.7904
0.5	116.9325	233.8649	23.6895
0.2	111.3596	556.798	27.4570
0.1	105.7950	1057.954	30.2446
0.01	80.3968	8039.678	39.0524



Figure 1.



INFOPLT 2.

Figure 2.



Figure 3.



Figure 4.



Figure 5.


3000.00
2419.98
1839.96
1259.94
679.92
99.90
FLOWRATE CM/SEC
0
.1
.2
.3
.4
.5
.6
.7
.8
.9
1.0
Z, AXIAL DISTANCE, SCALED

Figure 6.



.30
.26
.22
.18
.14
.10
DENSITY, KG/M**3
0
.1
.2
.3
.4
.5
.6
.7
.8
.9
1.0
Z, AXIAL DISTANCE, SCALED

Figure 7.


15.10
15.04
14.98
14.92
14.86
14.80
PRESSURE, TORR
0
.1
.2
.3
.4
.5
.6
.7
.8
.9
1.0
Z, AXIAL DISTANCE, SCALED

Figure 8.


Z, AXIAL DISTANCE, SCALED
0
.1
.2
.3
.4
.5
.6
.7
.8
.9
1.0
TEMPERATURE, DEG C
26.80
26.86
26.92
26.98
27.04
27.10

Figure 9.